New Explorations of Old Mathematics via Spreadsheets

Deane Arganbright
Divine Word University, Papua New Guinea
argandeane@yahoo.com

Abstract
Over the years, many once-common traditional topics and techniques have largely disappeared from the everyday mathematics cosmos. Surprisingly, exploring now-forgotten mathematics through the creative use of spreadsheets can provide new, challenging, and interesting teaching and learning experiences. The projects considered here provide glimpses of historical ideas, while showing the underlying mathematics of algorithms that is now often hidden within powerful computational tools. We present interactive models of such varied topics as counting systems, calculating procedures and devices, graphing techniques, and algorithms that incorporate both animated graphics and effective visual presentations using the spreadsheet display itself. Examples are drawn from such diverse undergraduate areas as geometry, calculus, linear algebra, differential equations, numerical analysis, group theory, probability, and statistics.

While the foundations of mathematics have generally remained unchanged over the years, the applications, algorithms, techniques, curricula, breadth, and foci of mathematics and mathematics education have undergone dramatic and rapid changes and expansion. In the process, the development of technology has left many previously key topics of the curriculum either languishing or forgotten. At the same time, we now can use the new technologies to re-examine old ideas, thus both preserving some of our mathematical history, and giving us new ways to develop our modeling skills. In this paper we use the electronic spreadsheet, epitomized by *Microsoft Excel*©, to take a fresh look at a variety of the old ideas, techniques, and algorithms of mathematics. Some of our examples still remain in the curriculum in various guises, but others disappeared long ago. Nonetheless, each of these can serve to whet our appreciation of the mathematics involved.

Our examples fall into two categories. First, we present modern models of older computations or algorithms in which we primarily try to employ the most basic mathematics operations, that are similar to those done by hand. Other examples illustrate ways to use *Excel*'s graphics to create visualizations of traditional concepts.

We anticipate that our examples will demonstrate a broad range of approaches that we can use to study historical mathematics, and will help teachers in designing spreadsheet models as they create their own classroom illustrations. The author will provide the underlying *Excel* files to those requesting them. We use *Excel 2010* to create our examples.

1. Counting and Arithmetic

In this initial section we look at ways to use *Excel* creatively to examine two of the historical building blocks of mathematics, counting and basic arithmetic.

Base 3 Counting. The counting system most familiar to us uses base 10, while computer operations typically employ base 2. To illustrate the nature of using a specific base in counting, for our first example in Figure 1, we employ base 3, using *Excel* graphics to illustrate the grouping process as we count 1, 2, 3, …, N. We employ a spin button to animate our model. We begin with a few snapshots of the model's output. With each click of the button, we increase the count, N, by 1 and show one more green dot in the range 1,2,3, …, 27 = 3^3. In addition, we indicate groups of 3 by blue rectangles, groups of three 3's by red ones, and groups of 27 by a black one.



Figure 1. Base 3 Counting Idea and Model Output

We create the locations for the dots and rectangles in four series of (x,y) values, using Cell Y1 for the value of our counter, N. We then use =IF() functions to form four corresponding series in creating an xy-graph of the appropriate objects for those points corresponding to $k \leq N$. We animate our graphic display by using a spin button to vary N. The Row 1 formulas in Figure 2 generate the digits in the binary expansion of N.

	X	Y	Z	AA	AB	AC	AD
1	N	25		0	2	2	1

	X	Y	Z	AA	AB	AC	AD
1	N	25		=MOD(INT($Y$1/27),3)	=MOD(INT($Y$1/9),3)	=MOD(INT($Y$1/3),3)	=MOD(Y1,3)

Figure 2. Base 3 Formulas

Binary Counting. We created the previous model as a means of visualizing the concept of computing in a given small base. There are many other ways to perform the computations themselves (see [11]). In Figure 3 we show one way to generate the binary (base 2) expansion as a text expression through the use of concatenation.

We enter the value of n, here n = 37, in Cell B1, use the =MOD() function in Cell B2 to find the final digit in the binary expansion, as the remainder when n is divided by 2 and form its text equivalent in Cell B3. We create the resulting integer part of the quotient in Cell C1 and the next digit in Cell C2. We use concatenation in Cell C3 to attach that next digit to the front of then the previous ones. Finally, we copy the formulas in Column C to the right to complete the rows.

	A	B	C	D	E	F	G
1	n	37	18	9	4	2	1
2		1	0	1	0	0	1
3		1	01	101	0101	00101	100101

	A	B	C	D
1	n	37	=(B1-B2)/2	=(C1-C2)/2
2		=MOD(B1,2)	=MOD(C1,2)	=MOD(D1,2)
3		=TEXT(B2,0)	=TEXT(C2,0)&B3	=TEXT(D2,0)&C3

Figure 3. Binary Counting

We also can create a graph like the one in Figure 4 to illustrate the process of Figure 3 while showing the resulting digits of the binary expansion.



Figure 4. Alternate Binary Model Display

Counting Using Cycles. The nation of Papua New Guinea, with a population of about 7.5 million, has over 800 distinct languages and hundreds of varied traditional counting systems. While many in both categories are in danger of disappearing from the national memory, the counting systems seem to being lost more quickly. Thadreina Abady [1] describes the use of *Excel* to demonstrate several of these systems. She gave a presentation on her work at ATCM 2011. Here we give a brief indication of one of them, the Aruamu language (about 40,000 speakers). In Figure 5 we have used a western notation for numerals. This is not a traditional notation, as most traditional languages were only communicated orally. Our chart is produced in *Excel*, using mathematical concepts (iteration) and concatenation.

	Mekesem	Pom^{e}ni
1	Wamra	1
2	Pom^{e}ni	2
3	Pom^{e}ni ko Mekesem	2+1
4	Pom^{e}ni ko Pom^{e}ni	2+2
5	Pom^{e}ni ko Pom^{e}ni ko Mekesem	2+2+1
6	Pom^{e}ni ko Pom^{e}ni ko Pom^{e}ni	2+2+2

Figure 5. Aruamu Counting

Extensive data on the diverse counting systems of Papua New Guinea were gathered by the late mathematician, Glendon Lean, and published in a 20-volume publication [10]. Lean describes the systems in terms of cycles, frames, and patterns as they are not true base *b* systems. Various systems usually group with 2, 5, 10, or 20 cycles. Some systems of PNG count using body parts.

Basic Arithmetic. Nowadays, most people resort to calculators to perform even the simplest arithmetical calculations. In Figure 6 we review the old pencil-and-paper operations, using standard U. S. layouts and a digit-by-digit approach. It should be recognized that there are variations in presentation styles among different national cultures.

For the operation of addition, in the following example we use a "carrying" technique. Thus, in last column, we add 9 + 2 + 8 to get 19. In the right digit of the answer, we generate 9 using mod 10. To get the 1 to carry into the column to the left, we subtract the 9 from the 19, and divide by 10 and form the resulting value (1) at top of column to left. We then repeat the process by copying the indicated formulas in Cells E8 and D2 to the left. Readers are encouraged to develop their own ways of implementing both addition and subtraction operations.

	A	B	C	D	E
1	Addition				
2	2	1	2	1	
3		6	6	7	9
4		8	5	9	2
5		6	4	8	8
6	2	1	7	5	9

	C	D	E
1		Addition	
2	=(SUM(D2:D5)-D6)/10	=(SUM(E2:E5)-E6)/10	
3	6	7	9
4	5	9	2
5	4	8	8
6	=MOD(SUM(C2:C5),10)	=MOD(SUM(D2:D5),10)	=MOD(SUM(E2:E5),10)

Figure 6. Addition

Figure 7 shows one model for multiplying an integer by a single digit number. We get the first digit in Cell E4 and its "carry" in Cell D1. We complete the model by copying these formulas to the left. Readers can develop alternative multiplication techniques, or create *Excel* models for division.

	A	B	C	D	E
1	2	5	3	6	
2		3	8	4	9
3	x				7
4	2	6	9	4	3

	D	E
1	=($E3*E2+E1-E4)/10	
2	4	9
3		7
4	=MOD($E3*D2+D1,10)	=MOD($E3*E2+E1,10)

Figure 7. Multiplication I

Multiplying by a many-digit number can be more daunting since it involves much additional carrying that we typically would do mentally when working by hand. We must either supply locations for each of these operations or build more complex formulas. The model of Figure 8 incorporates both approaches. Details are found in the files available from the author.

	A	B	C	D	E	F	G	H
1					9	5	2	6
2				x	2	7	4	6
3				5	7	1	5	6
4			3	8	1	0	4	
5		6	6	6	8	2		
6	1	9	0	5	2			
7	2	6	1	5	8	3	9	6
8	1	1	2	1	0	0		

Figure 8. Multiplication II

In Figure 9 we see the output of two other multiplication approaches. Below we provide the basic formulas used.

	A	B	C	D	E	F	G	H
1	Multiplication - 1							
2					8	5	9	2
3				x	6	4	7	8
4					64	40	72	16
5				56	35	63	14	
6			32	20	36	8		
7		48	30	54	12			
8	5	55	76	145	158	119	87	16
9	5	5	6	5	8	9	7	6

	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	X
1	Multiplication - 2														
2					8	5	9	2							
3				x	6	4	7	8							
4				6	8	7	3	6				6	4	7	1
5			6	0	1	4	4				6	4	6	1	
6		3	4	3	6	8				3	2	3	0		
7	5	1	5	5	2				5	3	5	1			
8	5	5	6	5	8	9	7	6							
9	0	0	1	1	1	1	0	0							

Figure 9. Multiplication III

G8: =SUM(G4:G7)+(H8-H9)/10, G9: =MOD(G8,10)
U4: =($Q3*N2+V4-N4)/10, T5: =($P3*N2+U5-M5)/10,
S6: =($O3*N2-L6+T6)/10, R7: =($N3*N2-K7+S7)/10

Peasant Multiplication. Books on the history of mathematics [6] discuss a variety of older alternative techniques for multiplication. In Figure 10 we use *Excel* to implement one of these, sometimes called peasant multiplication [2], [12]. It also shows the implicit use of the binary representation of an integer. To multiply two integers, in Column A we repeatedly divide the first number by 2, dropping any remainder. At same time, in Column B we repeatedly double the second one. We continue until the entry in Column A reaches 1. The desired product is then the sum of the integers in B that correspond to the odd integers in Column A. Note that what results is

$$105\cdot197 = (1\cdot2^0 + 0\cdot2^1 + 0\cdot2^2 + 1\cdot2^3 + 0\cdot2^4 + 1\cdot2^5 + 1\cdot2^6) = 1\cdot1 + 1\cdot8 + 1\cdot32 + 1\cdot64)\cdot197$$
$$=197 + 1,576 + 6,304 + 12,608 = 20,685$$

and that the binary expansion of 105 is 1101001_2. An advantage for users of this approach was that they needed only to double, halve, and add integers to do multiplication. The multiplication by 1 or

0 in Column D can also be done using a basic logic approach, e.g. D5: =IF(C5=1,B5,0). Further computational schemes to implement in a spreadsheet any be found in [11].

	A	B	C	D
1	Peasant Multiplication			
2	Small	Large		*Check*
3	105	197		*20,685*
4	A	B	Odd	Value
5	105	197	1	197
6	52	394	0	0
7	26	788	0	0
8	13	1576	1	1,576
9	6	3152	0	0
10	3	6304	1	6,304
11	1	12608	1	12,608
12			Product	20,685

	A	B	C	D
1	Peasant Multiplication			
2	Small	Large		*Check*
3	105	197		*=A3*B3*
4	A	B	Odd	Value
5	=A3	=B3	=MOD(A5,2)	=B5*C5
6	=INT(A5/2)	=2*B5	=MOD(A6,2)	=B6*C6
7	=INT(A6/2)	=2*B6	=MOD(A7,2)	=B7*C7
8	=INT(A7/2)	=2*B7	=MOD(A8,2)	=B8*C8
9	=INT(A8/2)	=2*B8	=MOD(A9,2)	=B9*C9
10	=INT(A9/2)	=2*B9	=MOD(A10,2)	=B10*C10
11	=INT(A10/2)	=2*B10	=MOD(A11,2)	=B11*C11
12			Product	=SUM(D5:D11)

Figure 10. Peasant Multiplication

2. Earlier Computation Tools and Methods

Within the brief period of the last two generations, many of the ways in which we carried out computations in the past have disappeared as markedly improved and powerful computational tools have been developed. Here we take a look at how we can use *Excel* to simulate some of the past procedures, devices, and procedures.

Logarithms. In earlier times, the multiplication and division of large numbers presented difficulties. Mathematicians sought to find more efficient ways to do such computations. One involved the use of logarithms. If we can write the numbers a and b as powers of 10, say $a = 10^c$ and $b = 10^d$, then multiplying b by a would become a matter of adding exponents: $ab = 10^c 10^d = 10^{c+d}$. The exponents of 10 are called base 10 logarithms. Thus if $a = 10^c$, then $c = \log_{10}(a)$. Eventually the values of logarithms and their inverses (called antilogarithms) were computed and appeared in printed tables. These tables were still in common use 50 years ago. We summarize the use of logarithms for multiplication in the *Excel* display of Figure 11.

	product		powers	number		log
	3462	→	$10^{3.539327064}$	3462	→	3.5393
x	539	→	x $10^{2.731588765}$	539	→	2.7316
			$10^{3.539327064 + 2.731588765}$	product		sum
	1,866,018	←	$10^{6.270915829}$	1,866,000	←	6.2709

Figure 11. Base 10 Logarithm Concept

Figure 12 shows a layout using the built-in =LOG10() function. We can also use this function to create a typical table of logarithms and then use table lookup functions – employing extrapolation and decimal and rounding conventions – to more fully demonstrate the old methods. Using *Excel* and logs to carry out division, exponentiation, and finding roots also provide good exercises.

	A	B
1	numbers	log
2	72.842	1.862382
3	381.54	2.58154
4	product	sum
5	27792.14	4.443922

	A	B
1	numbers	log
2	72.842	=LOG10(A2)
3	381.54	=LOG10(A3)
4	product	sum
5	=10^B5	=B2+B3

Figure 12. Base 10 Logarithms in *Excel*

Slide Rule. A subsequent step in using logarithms came through the use of a slide rule. The idea of a slide rule is shown below. If we have two sticks measured in equal subdivisions from 0.0 to 1.0, then we find 0.3 + 0.2 = 0.5 as shown in Figure 13. If we think of the values, c, as logarithms and place the corresponding values 10^c on the sticks, then our stick adding process finds $2.00 \times 1.58 = 10^{0.3} \cdot 10^{0.2} = 10^{0.3+0.2} = 10^{0.5} = 3.16$.



Figure 13. Multiplying using Logarithms

This procedure is the basis of the slide rule that was used by students, engineers, and scientists until the development of calculators and computers. Perhaps their major drawback was that they could produce only 3 significant digits of accuracy. Nonetheless, this topic can become an eye-catching artifact to use in the classroom.

An associated file provides an *Excel* simulation version, as illustrated in Figure 14. We use scroll bars to move the center piece and the cross-hair slider.



Figure 14. Animated Slide Rule

Napier's Rods. The mathematician John Napier developed many logarithm concepts [6]. He also created another physical means to carry out multiplication problems using addition as shown in Figure 15. He used rods, or bones, consisting of the single digit multiples of the digits (the 6-rod is at the left), and used them to carry out multiplication (at the right).

Unlike Napier, we include 0 in our model. By arranging the rods as shown, we can read off the multiples of one of the numbers and then add them. The need to carry can complicate the resulting design, either by using additional columns (hidden here) or by employing more complex formulas.

Figure 16 shows the spreadsheet layout, and the methods of obtaining products from the display. We use the Format, Borders and the Merge and Center features of *Excel* in creating our design.

The right display illustrates the carrying obstacle that we need to consider.



Figure 15. Napier's Rods Multiplication



Figure 16. Explanation of Rod Operation

3. Some Algorithms of Yore

Not only have modern technological devices essentially eliminated the use of once common computational techniques, but they also have served to reduce the importance of some previously widely-used algorithms. Once again, we show how we can use *Excel* to implement some of these techniques with new vitality.

Euclid's GCD Algorithm. Today's software tools make finding the greatest common divisor (gcd) of two integers a trivial task. In *Excel* we have the built-in =GCD() function. Euclid's gcd algorithm shows us one traditional way to obtain the result ourselves and provides us with additional useful information. The algorithm is based on the fact that if we have two positive integers a and b, with $a > b$, then we can divide a by b, getting a remainder $r_1 \geq 0$, and write $a = mb + r_1$, with $0 \leq r_1 < b$. In addition, if an integer n divides both a and b, then it also divides r_1. If $r_1 > 0$, we replace a by b and b by r_1, and repeat the process, continuing until we obtain a remainder of 0. When we obtain 0 as the remainder, the remainder immediately prior to 0 is the desired gcd [11]. We carry out this process in the first four columns of Figure 17.

	A	B	C	D	E	F	G	H	I	J
1	356	252	GCD:	4	= 17*356-24*252				17	-24
2					1	0				
3	big	little	quo	rem	0	1				
4	356	252	1	104	1	-1				
5	252	104	2	44	-2	3				
6	104	44	2	16	5	-7				
7	44	16	2	12	-12	17				
8	16	12	1	4	17	-24		4	17	-24
9	12	4	3	0	-63	89				
10	4	0	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!				
11	0	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!				
12	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!				

	A	B	C	D
1	356	252		
2				
3	big	little	quo	rem
4	=A1	=B1	=INT(A4/B4)	=A4-B4*C4
5	=B4	=D4	=INT(A5/B5)	=A5-B5*C5
6	=B5	=D5	=INT(A6/B6)	=A6-B6*C6
7	=B6	=D6	=INT(A7/B7)	=A7-B7*C7
8	=B7	=D7	=INT(A8/B8)	=A8-B8*C8
9	=B8	=D8	=INT(A9/B9)	=A9-B9*C9
10	=B9	=D9	=INT(A10/B10)	=A10-B10*C10
11	=B10	=D10	=INT(A11/B11)	=A11-B11*C11
12	=B11	=D11	=INT(A12/B12)	=A12-B12*C12

Figure 17. Euclid's Method for GCD

Using algebra, we can work backwards through our results to find integers s and t such that $\gcd(a,b) = sa + tq$. This result is used in abstract algebra. Discovering a pattern for obtaining s and t directly from our *Excel* layout makes for a challenging exercise. We have implemented this iteratively in our computations in Columns E:F using $s_n = s_{n-2} - s_{n-1}q_n$ and $t_n = t_{n-2} - t_{n-1}q_n$. The gcd and the desired values s and t come from the row containing the last non-zero remainder. Our model demonstrates a means for transferring these values via formulas.

We can also find (possibly different) values for s and t by using *Excel*'s Solver as in Cells E1:M1 of Figure 18. We enter C1: =GCD(A1,B1), M1: =E1*A1+I1*B1. We also enter =C1-M1 into Cell M2 as the solver goal that we wish to drive to zero. We then issue the command: Data, Solver, and enter M2 as the objective cell, with 0 as its value, by changing Cells E1, I1. We must also include constraints that require both E1 and I1 to be integers. We do not want to have a tick in the box for making the constrained variables positive. We then press the Solve button to obtain values for s and t. Note that the values of s and t are not unique.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	356	252	4			*	356	+		*	252	=	0
2	a	b	gcd								solver goal		4

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	356	252	4		-46	*	356	+	65	*	252	=	4
2	a	b	gcd								solver goal		0

Figure 18. GCD as Linear Combination via Solver

Unit Fractions. Early Egyptian mathematicians made use of unit fractions of the form $1/n$, where n is a positive integer. It can be shown that any common fraction, m/n, where $m < n$, can be written (generally in many ways) as the sum of distinct unit fractions. In Figure 19 we look at one such algorithm. It falls into the general category of "greedy" algorithms, where we repeatedly choses the largest denominator possible.

To see the approach, suppose that we look at the fraction 19/78. What is the largest positive integer n_1 for which $1/n_1 < 19/78$? We could use trial-and-error to find n_1. In doing so, we would find that $1/5 < 19/78$, while $1/4 > 19/78$. Thus, we would use $n_1 = 1/5$ as our first unit fraction, and then find the largest integer, n_2, for which $1/n_2 < 19/78 – 1/5 = 17/390$, and continue the process.

However, with a little thought, we can do better than trial-and-error. Since 1/5 is the largest unit fraction that is less than 19/78. then 5 is the smallest integer greater than or equal to 78/19. So if we find n = INT(78/19) + 1, that is smallest integer greater than 78/19, and $1/n$ will the largest unit integer less than 78/19. That is the basis for our algorithm, which uses the greatest integer function, =INT(). However, we need to consider the case when the difference is already a unit integer. For example if m/n = ¾, then k = ¾ – ½ = ¼ which is a unit fraction, so we should write ¾ = ½ + ¼ and quit. However, if we took =INT(4/1)+1 = 5, we would continue 1/2+1/5 + ….

Our model appears below. We also can use *Excel* to check that we get the correct answer. Designing a check to use integer arithmetic for this algorithm or for adding factions in general make good exercises. We can use *Excel*'s greatest common denominator, =GCD() and least common multiple =LCM() functions in the design as a check.

	A	B	C	D
1	num	den	r	fract
2	19	78	5	1/5
3	17	390	23	1/23
4	1	8970	8970	1/8970
5	0	8E+07	#DIV/0!	#DIV/0!

	A	B	C	D
1	num	den	r	fract
2	19	78	=IF(B2/A2=INT(B2/A2),B2/A2,INT(B2/A2)+1)	="1/"&TEXT(C2,0)
3	=(A2*C2-B2)	=B2*C2	=IF(B3/A3=INT(B3/A3),B3/A3,INT(B3/A3)+1)	="1/"&TEXT(C3,0)
4	=(A3*C3-B3)	=B3*C3	=IF(B4/A4=INT(B4/A4),B4/A4,INT(B4/A4)+1)	="1/"&TEXT(C4,0)
5	=(A4*C4-B4)	=B4*C4	=IF(B5/A5=INT(B5/A5),B5/A5,INT(B5/A5)+1)	="1/"&TEXT(C5,0)

Figure 19. Unit Fraction Model

Cramer's Rule. In linear algebra we typically encounter both determinants and Cramer's Rule, which uses determinants to solve $n \times n$ systems of linear equations. However, this method is unwieldy to use, especially for $n > 3$. Surprisingly, employing *Excel*'s built-in determinant function, =MDETERM(), we can implement Cramer's rule for larger systems. Here we illustrate our procedure using 5×5 matrices. Details of Cramer's Rule are found in a standard linear algebra book [9]. The rule determines each component as $x_n = |A_n|/|A|$ where $|A|$ is the determinant of the coefficient matrix A, and $|A_n|$ is the determinant of the matrix obtained by replacing the n^{th} column of A by the column of constants. Cramer's rule applies only when $|A| \neq 0$.

We present our design in Figure 20. Cell E5 contains the number, n, of one of the columns. The formulas for A_n in Columns H:L use the value of n in Cell E5. We enter =IF(A$3=$E$1,$F4,A4) in Cell H4 and then copy it throughout the block. In Cell B2, =MDETERM(A4:E8) computes the determinant $|A|$, while we find $|A_n|$ in Cell J1: =MDETERM(H4:L8).

Next, we use *Excel*'s Data Table feature in the Block O3:P8. We initially leave the top row of the block empty and generate the list of n values down the first column. Then in Cell P3 we enter the formula =J1/B1 which computes the value of the current n^{th} solution component. Next, we use the mouse to highlight the block O3:P8 and issue the command: Data, What If Analysis, Data Table. In the ensuing dialog box we click in the Column Input Cell box, followed by clicking on Cell E1 to provide the value of n. When we then click the OK button, *Excel* repeatedly enters each of the values of n listed in Column O into Cell E3, recording the resulting values for $|A|/|A_n|$ in the corresponding entries Column P. In Figure 20b we check our answers using *Excel*'s Matrix functions to find the solutions in Column Q as $A^{-1}C$.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q
1	|A|	160		n	2			$\|A_n\|$		-320				data table			ck
2	coefficient matrix A							matrix A_n							n	x_n	x_n
3	1	2	3	4	5	con										-2	
4	3	2	1	8	-3	13		3	13	1	8	-3		x_1	1		
5	5	-2	1	1	-3	22		5	22	1	1	-3		x_2	2		
6	2	0	2	-3	-1	9		2	9	2	-3	-1		x_3	3		
7	1	1	-1	2	3	7		1	7	-1	2	3		x_4	4		
8	-1	3	1	0	3	-1		-1	-1	1	0	3		x_5	5		

	M	N	O	P	Q
1	data table				ck
2			n	x_n	x_n
3				-2	
4		x_1	1	4	4
5		x_2	2	-2	-2
6		x_3	3	3	3
7		x_4	4	1	1
8		x_5	5	2	2

Figure 20. Cramer's Rule with Check

Old Square Root Methods. Today we simply click on a button in any number of electronic devises in order to find the square root of a number that will be correct to many decimal places. Using *Excel* we can use simple trial-and-error to think through estimations as we indicate in Figure 21.

	A	B	C	D
1	Find square root of n:			666
2	Estimate:	28	28	784

	A	B	C	D
1	Find square root of n:			666
2	Estimate:	25.8	25.8	665.64

	A	B	C	D
1	Find square root of n:			666
2	Estimate:	25.805	25.805	665.9

Figure 21. Square Root Using Trial-and-Error

However, we next look at a very complicated square root algorithm that was taught in the author's high school algebra class 60 years ago. Here we provide only a brief introduction using Figure 22. Because of its complexity, we refer readers to an explanation provided on the Web [7].

We first group the digits in the number into groups of two, both before and after a decimal point. Thus we would write 06 45 00 00. In our *Excel* displays, however, some zeroes will be dropped. We first find the largest integer (2) whose square (4) is less than the first group (6). We put the 2 on the top line and double the 2 to get 4 below the 6, subtracting to get 2. We then bring down the next group to produce 245. Next we double the 2 to get 4 and use trial-and-error to find the largest digit k (here 5) for which 25×45 = 225 that is less than or equal to 245. We then subtract and repeat the process.

This is certainly a non-intuitive and very complex process. We show several steps in Figure 22 as implemented in an *Excel* model.



Figure 22. An Old Traditional Square Root Computation

Square Roots and Newton's Method. There is another, and now better-known, old method for finding square roots that is very easy to implement on a spreadsheet. Theoretically, the work could be done by hand, but it would involve very lengthy divisions of long decimal expansions.

We first make a good estimate, say x_0, of the square root of a real number a. If we are correct, then $x_0 \cdot x_0 = a$, and $x_0 = a/x_0$. However, unless we are lucky, either x_0 is too large (so that a/x_0 is too small) or x_0 is too small (so that a/x_0 is too large). In either case the average (*i.e.* the arithmetic mean) of the two, x_0 and a/x_0, should be a better approximation. So we use this as the next estimate, and then we keep repeating the process [2], [11].

In Figure 23, we enter the value for the initial estimate, x_0, in Cell B3 and compute a/x_0 in Cell C3. We then create our next approximation, x_1, in Cell B4 as (B3+C3)/2 and copy the formulas down their respective, noting a rapid convergence, even with a poor first approximation.

	A	B	C
1	a	645	
2	n	x_n	a/x_n
3	0	20	32.25
4	1	26.125	
5	2		

	A	B	C
1	a	645	
2	n	x_n	a/x_n
3	0	20	32.25
4	1	26.125	24.688995
5	2	25.406998	25.386707
6	3	25.396852	25.396848
7	4	25.39685	25.39685
8	5	25.39685	25.39685

	A	B	C
1	a	645	
2	n	x_n	a/x_n
3	0	20	=$B$1/B3
4	=1+A3	=(B3+C3)/2	=$B$1/B4
5	=1+A4	=(B4+C4)/2	=$B$1/B5
6	=1+A5	=(B5+C5)/2	=$B$1/B6
7	=1+A6	=(B6+C6)/2	=$B$1/B7
8	=1+A7	=(B7+C7)/2	=$B$1/B8

Figure 23. Square Root Using Newton's Method

This algorithm is a special case of Newton's Method [4], a classical algorithm from calculus in which we compute an approximation of a zero of a differentiable function $y = f(x)$ by taking an initial estimate x_0, and then computing subsequent estimates by $x_{n+1} = x_n - f(x_n)/f'(x_n)$.

4. Visualizations of Classical Mathematics

One of *Excel*'s great strengths for communicating mathematics lies in the manner in which we can create interactive and animated graphics. Here we illustrate this through examples coming from various fields of mathematics.

Means. There are a great number of ways to define the mean of two numbers, a and b [6]. Perhaps the most familiar one is the *arithmetic* mean, $x = (a+b)/2$. As shown in Figure 24, this is the point that lies midway between a and b. Here $x–a = b–x$, so that $2x = a+b$, and $x = (a+b)/2$. Another, the *geometric* mean, is located so that the ratios of the distance from a to x and x to b are equal. Thus, $x/a = b/x$, $x^2 = ab$, and $x = \sqrt{ab}$. Others include: *harmonic*: $x = 2ab/(a+b)$; *heronian*: $x = a+\sqrt{ab}+b$; *contraharmonic*: $x = (a^2+b^2)/(a+b)$; *root-mean-square*: $x = \sqrt{a^2 + b^2}$; *centroidal*: $x = 2(a^2+ab+b^2)/3(a + b)$.



Figure 24. Arithmetic and Geometric Means

There are a variety of ways to exhibit these means, such as in the interactive *Excel* graph of Figure 25. As we vary a and b, our graph is updated automatically, showing the order indicated. To show mathematically that the sizes are always in the order shown is a good exercise.



Figure 25. Relations among Means

An exercise from Eves [6] uses a trapezoid to illustrate the different means. When shown in this way, we can visualize some of the properties that also provide us with good exercises. For example, the root-mean-square divides the trapezoid into two areas of equal area, and the geometric mean divides the trapezoid into two similar trapezoids.



Figure 26. Means Seen Through Trapezoids

Cycloids. As a unit circle rolls along the x-axis, the point that originally lies on the bottom at the origin traces out a curve called a cycloid. This curve has generated great interest among mathematicians for centuries [3], [6]. There are several ways to produce this curve. One is to use the parametric equations of the curve. Our animated model creates a graph (Figure 27a) to helps us to determine the curve's parametric equations: $x = t - \sin t$, $y = 1 - \cos t$. However, a second model (Figure 27b) uses *Excel*'s data table to create the curve. This approach is useful in constructions when it may be difficult to find the equations directly. In addition, our model simultaneously implements the trapezoidal rule [4] from calculus to show that the area under one arch of the cycloid is equal to the sum of the areas of 3 unit circles. We can verify this directly by using integration from calculus. As an exercise, we can generalize this model for a circle of radius r. We use a scroll bar to generate the animation.



Figure 27. Animated Cycloid

Gaussian Pivoting. One of the fundamental topics studied in linear algebra is finding the solution of a system of linear equations. One way to do this is through the use of Gaussian pivoting. Below we see the output of a manual solution using elementary matrices in an *Excel* implementation. It is not always apparent to students how the term "pivoting" applies. Our animated model illustrates this for a 2×2 system..

In the left of Figure 28 we see the steps in eliminating the x term from the second equation. We do this by subtracting the multiple of 3 of the first equation from the second. However, if we gradually change the 3 to 0 in small steps by using a scroll bar (or slider), then we see the red line pivoting around the solution point. We then use a similar technique on the second component to pivot the blue line. Figure 29 shows the screen display and snapshots of its changing graph.

$$\begin{bmatrix} 0.5 & 0 \\ 0 & 1 \end{bmatrix}\left[\begin{array}{cc|c} 2 & 4 & 14 \\ 3 & 8 & 23 \end{array}\right] \quad \begin{array}{l} 2.00x + 4.00y = 14.00 \\ 3.00x + 8.00y = 23.00 \end{array} \qquad \begin{bmatrix} 1 & 0 \\ 0 & 0.5 \end{bmatrix}\left[\begin{array}{cc|c} 1 & 2 & 7 \\ 0 & 2 & 2 \end{array}\right] \quad \begin{array}{l} 1.00x + 2.00y = 7.00 \\ 0.00x + 2.00y = 2.00 \end{array}$$

$$\begin{bmatrix} 1 & 0 \\ -3 & 1 \end{bmatrix}\left[\begin{array}{cc|c} 1 & 2 & 7 \\ 3 & 8 & 23 \end{array}\right] \quad \begin{array}{l} 1.00x + 2.00y = 7.00 \\ 3.00x + 8.00y = 23.00 \end{array} \qquad \begin{bmatrix} 1 & -2 \\ 0 & 1 \end{bmatrix}\left[\begin{array}{cc|c} 1 & 2 & 7 \\ 0 & 1 & 1 \end{array}\right] \quad \begin{array}{l} 1.00x + 2.00y = 7.00 \\ 0.00x + 1.00y = 1.00 \end{array}$$

$$\left[\begin{array}{cc|c} 1 & 2 & 7 \\ 0 & 2 & 2 \end{array}\right] \quad \begin{array}{l} 1.00x + 2.00y = 7.00 \\ 0.00x + 2.00y = 2.00 \end{array} \qquad \left[\begin{array}{cc|c} 1 & 0 & 5 \\ 0 & 1 & 1 \end{array}\right] \quad \begin{array}{l} 1.00x + 0.00y = 5.00 \\ 0.00x + 1.00y = 1.00 \end{array}$$

Figure 28. Gaussian Pivots through Elementary Matrices



Figure 29. Output of Animated Pivoting in 2×2 Systems

A significant challenge will be to use a perspective drawing to create a similar visualization model for a 3×3 system.

Statistical Picture Charts. *Excel* is an excellent tool for creating animated graphs [3,11]. These are especially helpful in statistics. Using the model for Figure 30 we explore the correct depiction of images of relative sizes. In one model we use a scroll bar to vary the radius of the red circle of the figure. Users can experiment to see if they can tell visually when the area of the smaller read image is a given percentage, say 50%, of the area of the original circle. Another uses a 3-dimensional soft drink bottle. Similar images can be included in a map or a similar figure to demonstrate the relative sizes of such statistical items as sales, production, or population of various locales in the map.

Examples of visual perception displays abound in statistics books and the classical book by Huff [8].



Figure 30. Images proportional to Data

Group Theory. One area that currently appears to be largely unexplored is the use of a spreadsheet to create visualizations for abstract algebra. Carter [5] addresses visualization in group theory. In Figure 31 we use the =MOD() function to create a multiplication table for the integers mod 6. We then use conditional formatting to show the normal subgroup {0,3} in red, with its cosets in green and yellow. We then represent the corresponding factor group in the same format, allowing us to demonstrate that the factor group is isomorphic to the integers mod 3.

	0	3	1	4	2	5
0	0	3	1	4	2	5
3	3	0	4	1	5	2
1	1	4	2	5	3	0
4	4	1	5	2	0	3
2	2	5	3	0	4	1
5	5	2	0	3	1	4

	R	G	Y
R	R	G	Y
G	G	Y	R
Y	Y	R	G

	0	1	2
0	0	1	2
1	1	2	0
2	2	0	1

Figure 31. Group Theory

Numerical Analysis. The field of numerical analysis is rich with algorithms that we can implement naturally as interactive *Excel* models. In Figure 32 we illustrate Euler's Method [2], [4], [13], a traditional introductory means for approximating the solution of a differential equations initial value problem, $y' = f(x,y)$, $y(x_0) = y_0$. We use $f(x,y) = x + y$, $y_0 = 1$. We have chosen a problem that we can solve analytically as $y = 2e^x – 1 – x$ for $x \geq 0$, in order to show the method visually.

Using the initial point $w_0 = x_0 = 0$, $y_0 = 1$, we choose an x-step size, dx. We use the tangent line to the solution curve at x_0 to approximate the next point as $x_1 = x_0 + dx$, $y_1 \approx w_1 = y_0 + f(x_0, w_0)dx$. We then continually repeat the process, using w_i as an approximation for y_i, where $x_{i+1} = x_i + dx$, $w_{i+1} = w_i + f(x_i,w_i)dx$. Since the points (x_i,w_i) usually will not lie on the solution curve, we find that the approximation generally deteriorates as x increases.

In our model we use a scroll bar to vary the number of points, n. This allows us to see how the approximation improves with larger n and smaller dx. Ultimately we also will realize there is a need to find more efficient approaches, which we also can implement in *Excel*.

	A	B	C	D	E	F	G
1	x0	0	y0	1		Euler	
2	b	2	n	4		y' = x + y	
3	dx	0.5					
4		approx.				actual	
5	n	x	y(est)	y'		x	y(act)
6	0	0.00	1.00	1.00		0.00	1.00
7	1	0.50	1.50	2.00		0.01	1.01
8	2	1.00	2.50	3.50		0.02	1.02
9	3	1.50	4.25	5.75		0.03	1.03
10	4	2.00	7.13	9.13		0.04	1.04

	A	B	C	D
1	x0	0	y0	1
2	b	2	n	10
3	dx	=(B2-B1)/D2		
4		approx.		
5	n	x	y(est)	y'
6	0	=B1	=D1	=B6+C6
7	=1+A6	=B6+$B$3	=C6+D6*$B$3	=B7+C7
8	=1+A7	=B7+$B$3	=C7+D7*$B$3	=B8+C8
9	=1+A8	=B8+$B$3	=C8+D8*$B$3	=B9+C9
10	=1+A9	=B9+$B$3	=C9+D9*$B$3	=B10+C10

Figure 32. Euler's Method in Differential Equations

Our graph of Figure 33 shows the resulting approximations for n = 4 and n = 10. Readers are encouraged to create similar models for various other algorithms of numerical analysis or operations research.



Figure 33. Animated Euler's Method Output

References

[1] Abady, T. "Spreadsheet as an Innovative Tool for Traditional Counting in Papua New Guinea", *Contemporary PNG Studies: DWU Research Journal*, 20, May 2014, pp. 66-76.

[2] Arganbright, D. *Mathematical Applications of Electronic Spreadsheets*. McGraw-Hill, New York, 1985.

[3] Arganbright, D. *Practical Handbook of Spreadsheet Curves and Geometric Constructions*. CRC Press, Boca Raton, FL, 1993.

[4] Burden, R. and J. Faires, Numerical Analysis, 10th ed. Dengage, Boston,2015.

[5] Carter, N. *Visual Group Theory*. Mathematical Assn. of America, Washington DC, 2009.

[6] Eves, H. *An Introduction to the History of Mathematics, 6th ed*. Saunders, Fort Worth. 1990.

[7] Home School Math: http://www.homeschoolmath.net/teaching/square-root-algorithm.php

[8] Huff, D. *How to Lie with Statistics*. Norton, New York, 1954.

[9] Lay, D. *Linear Algebra and its Applications, 4th ed*. Pearson, Boston, 2012.

[10] Lean, G. *Counting Systems of Papua New Guinea, 2nd ed*. (20 volumes). Papua New Guinea Institute of Technology, Lae, PNG, 1991.

[11] Neuwirth, E. and D. Arganbright. *The Active Modeler: Mathematical Modeling with Microsoft Excel*. Cengage, Boston, 2004.

[12] Ore, O. *Number Theory and its History*. McGraw-Hill, New York, 1948.

[13] Zill, D. *A First Course in Differential Equations with Modeling Applications, 7th ed*. Cengage, Pacific Grove, CA, 2001.